

Eines informàtiques (MAPLE)

Pràctica 6

Matemàtiques per a ensenyants, curs 2002-2003.

6 Endomorfismes, vectors i valors propis, diagonalització...

Un dels problemes típics quan es treballa amb endomorfismes d'un espai vectorial de dimensió finita és el de determinar bases respecte de les que la matriu de l'endomorfisme és el més simple possible. Per a poder fer aquest estudi hi juga un paper fonamental la determinació dels vectors i valors propis i dels subespais associats. En el paquet **LinearAlgebra** de Maple es poden trobar funcions per a realitzar totes les operacions necessàries.

6.1 Polinomi característic i valors propis

Per a determinar el polinomi característic d'una matriu M podem realitzar els càlculs construint la matriu $M - x \text{Id}$ i demanant el determinat d'aquesta nova matriu

```
> with(LinearAlgebra):
> M:=< <1,1,3> | <2,7,-2> | <1,3,-6> >;
> Mx:= simplify(M-x*IdentityMatrix(3));
> de:=Determinant(Mx);
```

o demanant directament que Maple faci el càlcul i doni directament el polinomi amb la comanda **CharacteristicPolynomial**

```
> CharacteristicPolynomial(M,x);
```

(triar una cosa o l'altre només és una qüestió de gust personal ja que els càlculs que s'han de realitzar internament són essencialment els mateixos).

Pera determinar els valors propis (i la seva multiplicitat) caldrà determinar les arrels del polinomi característic. També es pot fer directament amb la comanda **solve** o amb la comanda específica del paquet **LinearAlgebra** que fa aquesta feina (**Eigenvalues**()).

```
> M1:=< <1,1,3> | <-1,7,-2> | <1,3,-6> >;
> solve(Determinant(M1-x*IdentityMatrix(3)));
> Eigenvalues(M1);
```

Com podreu veure, la diferència entre els dos resultats és ben poca.

6.2 Vectors propis

Un cop es coneixen els valors propis associats a una matriu A , la determinació dels espais de vectors $\vec{v}$ per als que es verifica alguna de les condicions $A\vec{v} = \lambda\vec{v}$ es redueix a la solució de sistemes d'equacions lineals homogenis. La comanda **NullSpace**() del paquet **LinearAlgebra** permet fer la feina.

Exemple 6.1

La matriu

$$A = \begin{pmatrix} -2 & 2 & -3 & -1 \\ 2 & -1 & 2 & 0 \\ 3 & -2 & 4 & 1 \\ -2 & 0 & -2 & -1 \end{pmatrix}$$

té per valors propis 1 i -1 tal i com es pot comprovar amb

```
> A:= <<-2|2|-3|-1>,<2|-1|2|0>,<3|-2|4|1>,<-2|0|-2|-1>>;
> Eigenvalues(A);
```

Els espais de vectors propis amb valor propi 1 o -1 es podran determinar amb

```
> B1:= A-IdentityMatrix(4);
> E1:=NullSpace(B1);
> B2:= A+IdentityMatrix(4);
> E2:=NullSpace(B2);
```

Podreu veure que $E1$ i $E2$ contenen cada una un parell de vectors que generen l'espai de vectors propis corresponent (multipliqueu A per cada una de les columnes que surten en $E1$ o $E2$ per a comprovar que es tenen, efectivament, vectors propis amb valor propi 1 o -1). A la vista dels resultats es pot afirmar que la matriu A és diagonalitzable i es pot confirmar aquest fet considerant

```
> E:=<op(E1)|op(E2)>;
> E^(-1).A.E;
```

Tot el procés de calcular els vectors propis associats a cada un dels valors propis d'una matriu té una comanda específica en el paquet **LinearAlgebra** que es crida amb **Eigenvectors()**. Si considereu

```
> EE:=Eigenvectors(A);
```

obtindreu una seqüència amb dos elements, el primer és una columna amb els valors propis (repetits tants cops com la multiplicitat) i el segon és una matriu que té per columnes les components dels vectors propis ordenades de la mateixa manera que els valors propis (si us hi fixeu, veureu que no surten exactament els mateixos vectors que quan apliqueu **NullSpace()**). Si feu la multiplicació

```
> EE[2]^(-1).A.EE[2];
```

tornareu obtenir la matriu diagonal equivalent a la matriu A .

Amb tot el que heu vist fins ara, l'única dificultat de l'exercici següent hauria de ser la introducció de la matriu M .

Exercici 6.1

Considereu la matriu

$$M = \begin{pmatrix} 1/2 & -1/2 & -1/2 & -1/2 & -1/2 & 0 & 1/2 \\ 7/2 & 9/2 & 3/2 & 7/2 & -3/2 & -1 & -1/2 \\ -1/2 & -5/2 & 1/2 & -1/2 & 3/2 & -2 & 1/2 \\ -5 & -4 & 0 & -6 & 6 & 4 & -1 \\ 0 & 1 & 1 & -2 & 3 & 3 & -1 \\ -7/2 & -9/2 & -1/2 & -11/2 & 11/2 & 3 & -1/2 \\ -3/2 & -3/2 & 5/2 & -11/2 & 17/2 & 4 & -3/2 \end{pmatrix}$$

- Calculeu el polinomi característic de M .
- Determineu els seus valors propis i les multiplicitats corresponents.
- Determineu els espais de vectors propis associats a cada un dels valors propis.
- Comproveu que la matriu M és diagonalitzable.
- Calculeu una matriu C tal que $C^{-1} \cdot M \cdot C$ sigui diagonal.

En aquest altre exercici podreu veure una matriu amb el mateix polinomi característic que la matriu M però que no és diagonalitzable.

Exercici 6.2

Considereu

$$B = \begin{pmatrix} 1/2 & -1/2 & -1/2 & -1/2 & -1/2 & 0 & 1/2 \\ 7/2 & 9/2 & 3/2 & 7/2 & -3/2 & -1 & -1/2 \\ -1/2 & -5/2 & 1/2 & -1/2 & 3/2 & -2 & 1/2 \\ -4 & -3 & 1 & -5 & 6 & 3 & -1 \\ 1 & 2 & 2 & -1 & 3 & 2 & -1 \\ -5/2 & -7/2 & 1/2 & -9/2 & 11/2 & 2 & -1/2 \\ 1/2 & 1/2 & 9/2 & -7/2 & 17/2 & 2 & -3/2 \end{pmatrix}$$

calculeu el seu polinomi característic i comproveu que no és diagonalitzable.

Una de les propietats bàsiques del polinomi característic d'una matriu és que *anul·la* a la matriu corresponent. És a dir, si $p(x)$ és el polinomi característic de la matriu A es compleix $p(A) = 0$. En l'exercici següent es posa de manifest aquest fet en un parell d'exemples.

Exercici 6.3

Considereu les matrius A donades per

$$\begin{pmatrix} 1 & 3 & 4 & -1 \\ 0 & 3 & 2 & 1 \\ -1 & -1 & 2 & 5 \\ 2 & 3 & 1 & -1 \end{pmatrix} \quad \begin{pmatrix} -2 & 3 & 1 & -1 \\ 2 & -1 & 0 & 1 \\ -1 & -2 & 2 & 3 \\ 3 & 3 & 1 & -1 \end{pmatrix}$$

calculeu els seus polinomis característics i comproveu que les anul·len.

D'entre tots els polinomis que anul·len un endomorfisme s'en pot determinar el que té el grau més petit (polinomi mínim). La comanda de Maple `MinimalPolynomial( )` calcula el polinomi mínim d'una matriu.

Exemple 6.2

Es pot calcular el polinomi mínim de

$$A = \begin{pmatrix} 0 & -1 & 2 \\ 3 & -4 & 6 \\ 2 & -2 & 3 \end{pmatrix}$$

amb

```
> A:= < <0,3,2>|<-1,-4,-2>|<2,6,3> >;
> pm:=MinimalPolynomial(A,x);
> Pm:=unapply(pm,x);
> Pm(A);
```

Observeu que la comanda `Pm` és la funció que s'obté *prenent com a variable* la `x` en l'expressió `pm`, que s'aplica correctament sobre una matriu i sobre A dóna com a resultat la matriu nul·la.

Exercici 6.4

Calculeu el polinomi característic de la matriu A anterior i comproveu que té els mateixos factors irreductibles que el polinomi mínim (només que amb multiplicitats menors).

6.3 Formes de Jordan

Com recordareu, un endomorfisme d'un espai vectorial (matriu quadrada) no sempre diagonalitza. Per a aquests casos és possible intentar obtenir altres *formes canòniques*. En l'exemple següent veureu una matriu 3×3 que té un únic valor propi i que, a més, l'espai de vectors propis només té dimensió 1.

Exercici 6.5

Sigui

$$A = \begin{pmatrix} 1 & 0 & 1 \\ -1 & 3 & 1 \\ 0 & -1 & 2 \end{pmatrix}.$$

- Calculeu el polinomi característic de A , comproveu que A té un únic valor propi λ (triple) i que l'espai de vectors propis de A té dimensió 1.
- Sigui $\vec{v}_3$ un vector propi no nul de A . Determineu un vector $\vec{v}_2$ tal que $(A - \lambda \cdot \text{Id})\vec{v}_2 = \vec{v}_3$ i un vector $\vec{v}_1$ tal que $(A - \lambda \cdot \text{Id})\vec{v}_1 = \vec{v}_2$.
- Comproveu que $\vec{v}_1, \vec{v}_2, \vec{v}_3$ són independents.
- Determineu la matriu $P^{-1} \cdot A \cdot P$ si P és la matriu que té per columnes les components dels vectors $\vec{v}_i$.

La matriu que haureu obtingut en l'últim apartat de l'exercici anterior és el que normalment s'anomena un *bloc de Jordan* (un mateix valor en tots els elements de la diagonal i 1 just en els llocs que estan per sobre, o per sota, d'aquesta diagonal). En general, es poden buscar formes canòniques de la forma $\left(\begin{array}{c|c|c} J_1 & 0 & \cdots \\ \hline 0 & J_2 & \cdots \\ \hline \vdots & \vdots & \ddots \end{array} \right)$, on cada un dels J_i és un bloc de Jordan.

De la mateixa forma que per als valors i vectors propis, dins del paquet **LinearAlgebra** hi ha comandes específiques per a determinar la forma de Jordan d'un endomorfisme i per a obtenir el canvi de base que permet obtenir aquesta forma a partir de qualsevol altra forma equivalent. La comanda que fa les dues feines és `JordanForm( )`. Per a obtenir la forma de Jordan equivalent a una matriu A donada basta fer `JordanForm(A)`; si es vol la matriu de canvi es pot modificar la comanda anterior i posar `JordanForm(A, output='Q')`; i si es volen les dues coses `JordanForm(A, output=['J', 'Q'])`. Veieu a continuació uns exemples

Exemple 6.3

```
> A:=<<2, 2, 0, 0>|<1/2, 3/2, -1/2, 1/2>|<0, 0, 2, 0>|<-1/2, -3/2, 1/2, 3/2>>;
> JordanForm(A);
> B:= JordanForm(A,output='Q');
> C:= JordanForm(A,output=['J','Q']);
> simplify(B^(-1).A.B);
> simplify(C[2]^(-1).A.C[2]);
```

No ha de ser difícil fer l'exercici següent:

Exercici 6.6

Considereu la matriu A donada per

$$\begin{pmatrix} 3 & 0 & 0 & 1 & 4 & -3 & 2 \\ -1 & 3 & 0 & -2 & -5 & 4 & -2 \\ 0 & 2 & 3 & -3 & -1 & 2 & 1 \\ 0 & 1 & 0 & 1 & -1 & 1 & 0 \\ -2 & -2 & -1 & 0 & -8 & 6 & -7 \\ -1 & -2 & -1 & 1 & -6 & 5 & -5 \\ 2 & 0 & 1 & 1 & 9 & -6 & 7 \end{pmatrix}$$

i la matriu B

$$\begin{pmatrix} 2 & 1 & -1 & 1 & 4 & -3 & 3 \\ 0 & 3 & 0 & -2 & -4 & 3 & -2 \\ 0 & 3 & 2 & -3 & 0 & 1 & 2 \\ 0 & 1 & 0 & 1 & -1 & 1 & 0 \\ 0 & -4 & 1 & 0 & -8 & 6 & -9 \\ 0 & -4 & 1 & 1 & -7 & 6 & -7 \\ 0 & 1 & 0 & 1 & 8 & -5 & 8 \end{pmatrix}$$

Existeix alguna matriu invertible P tal que $P^{-1} \cdot A \cdot P = B$? (Determineu les formes de Jordan de A i B).

Exercici 6.7

Comproveu que existeix una matriu Q (com a mínim) tal que $Q^{-1} \cdot A \cdot Q = B$ per a

$$A = \begin{pmatrix} 5/2 & -1/2 & -1/2 & 0 & 0 \\ -5 & -1 & -1 & 1 & -1 \\ 11/2 & 5/2 & 5/2 & -1 & 1 \\ 2 & 0 & 0 & 1 & 0 \\ 6 & 4 & 2 & -1 & 3 \end{pmatrix} \quad B = \begin{pmatrix} 1 & 0 & 0 & -1 & 0 \\ 1 & 1 & -1 & 1 & 1 \\ 3 & -1 & -1 & 2 & 3 \\ 0 & 1 & 1 & 2 & -1 \\ 3 & -2 & -3 & 2 & 5 \end{pmatrix}$$

i calculeu una d'aquestes matrius Q .